Maple-assisted proof of formula for A197237

Robert Israel

26 Sep 2018

There are $5^6 = 15625$ configurations for a 2×3 sub-array, but not all can arise.

We encode these configurations as lists in the order $\begin{bmatrix} x_1 & x_2 & x_3 \\ x_4 & x_5 & x_6 \end{bmatrix}$. Let $b_0 = 4$, $b_1 = 3$, $b_2 = 2$, $b_3 = 1$, $b_4 = 2$. If $x_i = j$, then the neighbours of site i in the sub-array that have value b_j must be either j or $j - 1$.

```
> b[0]:= 4: b[1]:= 3: b[2]:= 2: b[3]:= 1: b[4]:= 2:
  goodconfig:= proc(x) local t;
     t:= numboccur(b[x[1]], [x[2],x[4]]);
     if t < x[1]-1 or t > x[1] then return false fi;
     t:= numboccur(b[x[2]], [x[1],x[3],x[5]]);
     if t < x[2]-1 or t > x[2] then return false fi;
     t:= numboccur(b[x[3]], [x[2],x[6]]);
     if t < x[3]-1 or t > x[3] then return false fi;
     t:= numboccur(b[x[4]], [x[1],x[5]]);
     if t < x[4]-1 or t > x[4] then return false fi;
     t:= numboccur(b[x[5]], [x[2],x[4],x[6]]);
     if t < x[5]-1 or t > x[5] then return false fi;
     t:= numboccur(b[x[6]], [x[3],x[5]]);
     t >= x[6]-1 and t <= x[6]
  end proc:
  Configs:= select(goodconfig, [seq(convert(5^6+i,base,5)[1..6],i=
  0..5^6-1)]):
  nops(Configs);
```

$$403 \qquad (1)$$

There are 403 allowed configurations.

Consider the 403×403 transition matrix T with entries $T_{ij} = 1$ if the first two rows of a 3×3 sub-array could be in configuration i while the last two rows are in configuration j, and 0 otherwise. The following code computes it.

```
> Compatible:= proc(i,j) local Xi,Xj,k;
   Xi:= Configs[i]; Xj:= Configs[j];
   if Xi[4..6] <> Xj[1..3] then return 0 fi;
   if numboccur(b[Xi[4]],[Xi[1],Xi[5],Xj[4]]) <> Xi[4] then return
  0 fi;
   if numboccur(b[Xi[5]],[Xi[2],Xi[4],Xi[6],Xj[5]]) <> Xi[5] then
  return 0 fi;
   if numboccur(b[Xi[6]],[Xi[3],Xi[5],Xj[6]]) = Xi[6] then 1 else 0
  fi;
  end proc:
  T:= Matrix(403,403,Compatible):
```

Thus for $n \ge 2$ $\quad a(n) = \dfrac{u^T\, T^{n-2} v}{2}$ where u is a column vector with 1 for configurations whose first row could be a top row, 0 otherwise, and similarly v has 1 for configurations whose second row could be a bottom row.

```
> u:= Vector(403,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[1]], [x[2],x[4]])=x[1]
     and numboccur(b[x[2]],[x[1],x[3],x[5]]) = x[2]
     and numboccur(b[x[3]],[x[2],x[6]]) = x[3] then 1 else 0 fi
  end proc):
  v:= Vector(403,proc(i) local x; x:= Configs[i];
     if numboccur(b[x[4]], [x[1],x[5]]) = x[4]
     and numboccur(b[x[5]],[x[2],x[4],x[6]]) = x[5]
     and numboccur(b[x[6]],[x[3],x[5]]) = x[6] then 1 else 0 fi
   end proc) :
```

To check, here are the first few entries of our sequence (apart from a_1, which doesn't really fit the pattern, although it does work with the recurrence).

```
> Tv[0]:= v:
  for n from 1 to 32 do Tv[n]:= T . Tv[n-1] od:
> A:= [seq(u^%T . Tv[n],n=0..32)];
```

$A := [5, 15, 46, 156, 507, 1637, 5338, 17401, 56648, 184384, 600287, 1954546, 6363740, 20718710, 67455328, 219621081, 715042212, 2328028685, 7579574414, 24677521267, 80344901649, 261586345777, 851670870944, 2772863697333, 9027869180467, 29392869900112, 95697088678798, 311569874170026, 1014406894858665, 3302698476141187, 10752901304065119, 35009216638776687, 113982748937165611]$ **(2)**

Now here is the empirical recurrence formula. It says that $u^T T^n Q(T)\ v = 0$ for all nonnegative integers n, where Q is the following polynomial.

```
> n:= 'n': empirical:= a(n) = 2*a(n-1) +8*a(n-3) +11*a(n-4) +17*a
  (n-5) +11*a(n-6) +5*a(n-7) -6*a(n-8) -16*a(n-9) -15*a(n-10) -3*a
  (n-11) +a(n-13):
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(13-i),i=0.
  .13),t);
```

$Q := t \mapsto t^{13} - 2\,t^{12} - 8\,t^{10} - 11\,t^9 - 17\,t^8 - 11\,t^7 - 5\,t^6 + 6\,t^5 + 16\,t^4 + 15\,t^3 + 3\,t^2 - 1$ **(3)**

In fact, it turns out that $Q(T)\ v = 0$.

```
> Qv:= add(coeff(Q(t),t,j)*Tv[j],j=0..13):
```

We check that this is 0:

```
> (Qv)^%T . Qv;
```

$$0$$ **(4)**

This completes the proof.